MATEMÀTIQUES II
Escola Universitària d'Enginyeria Tècnica Industrial de Barcelona

UNIVERSITAT POLITÈCNICA DE CATALUNYA
BARCELONATECH

Francesc Pozo, Núria Parés, Yolanda Vidal
http://www.euetib1.upc.edu/gimel

Extremos relativos y condicionados

Objetivos:

- **En esta práctica se extienden las técnicas de estudio de los valores extremos para una función real de una variable a funciones de varias variables, estudiando tres tipos de extremos: relativos, absolutos y condicionados.**
- **Calcular los valores extremos relativos, puntos de silla, absolutos y condicionados de funciones de varias variables.**
- **Otras aplicaciones:**

  - Uso de comandos para el cálculo de máximos y mínimos: Jacobian, Hessian, Determinant y extrema.

  - Elaboración de programas para representar gráficamente funciones de varias variables.

  - Gráficas de máximos y mínimos.

Extremos relativos y puntos de silla

Para calcular los extremos relativos de una función escalar $f:\mathbb{R}^n \rightarrow \mathbb{R}$, el primer paso es calcular los puntos críticos, es decir, aquellos que anulan las primeras derivadas parciales. El vector de las primeras derivadas se denomina gradiente, que en este caso coincide con la matriz Jacobiana.

Con Maple, se puede calcular la matriz jacobiana de una función con el comando **Jacobian**. Para ello, es necesario primero cargar la libreria **Student[VectorCalculus].** Por medio del comando **solve** se encuentran los valores de x e y que anulan el Jacobiano.

Ejemplo: Calcular los puntos críticos de la función $f(x, y) = x^2 y^2$

```
> restart:
> with(Student[VectorCalculus]):
> f := (x, y)->x^2*y^2;
> J:=Jacobian([f(x,y)],[x,y]);
> ptos := solve({J[1, 1] = 0, J[1, 2] = 0}, {x, y});
```

Para decidir cuáles de estos puntos son posibles extremos, se evalua la segunda derivada en cada uno de esos puntos. La matriz que contiene todas las segundas derivadas de una función se denomina matriz Hessiana. Dependiendo del tipo de matriz resultante, se decide qué tipo de extremo tiene la función en ese punto:

- Si el determinante de la matrix hessiana (o hessiano) es igual a cero, entonces no tenemos ninguna información
- Si el hessiano es menor a cero, entonces el punto es un **punto de silla**
- Si el hessiano es mayor a cero, entonces el punto es un **extremo.** Para saber si es un máximo o un mínimo se estudian los signos de las segundas derivadas.

$\frac{\partial^2}{\partial x^2} f(x_c, y_c) > 0$, entonces el punto es un **mínimo**

$\frac{\partial^2}{\partial x^2} f(x_c, y_c) < 0$, entonces el punto es un **máximo**

Con Maple, la matriz Hessiana se calcula con el comando **Hessian** que también forma parte de la libreria **Student[VectorCalculus]**. Para calcular el determinante se utiliza el comando **Determinant** de la libreria **Student[LinearAlgebra]**, y para evaluarlo, el comando **eval**.

Ejemplo: Calcular el hessiano en cada uno de los puntos críticos de la función $f(x, y) = x^2 y^2$ y determine qué tipo de extremo es.

```
> H := Hessian(f(x, y), [x, y]);
> with(Student[LinearAlgebra]):
> eval(Determinant(H),ptos[1]);
> eval(Determinant(H),ptos[2]);
> eval(Determinant(H),ptos[3]);
```

¿Qué se puede deducir a partir de los resultados obtenidos?

Representemos gráficamente la función y los puntos para comprobar los resultados:

```
> with(plots):
```

```
> F:=plot3d(f(x,y),x=-10..10,y=-10..10):
> a1:=spacecurve([x,0,0],x=-10..10,color=blue, thickness=5):
> a2:=spacecurve([0,y,0],y=-10..10,color=red, thickness=5):
> display3d([F,a1,a2],axes=frame);
```

Con Maple es relativamente fácil encontar los extremos, ya que la instrucción **extrema** los calcula directamente. La instrucción **extrema(expr, constraints, vars, 's'),** calcula los extremos de la función **expr**, con las condiciones de ligadura definidas por **constraints**. Los candidatos a extremos o puntos críticos quedan almacenados en la variable **s**.

```
> extrema(f(x,y),{},{x,y},'s'):
> s;
```

Extremos absolutos en regiones limitadas

Para encontrar los valores extremos absolutos de una función en una región cerrada y acotada, es necesario tener en cuenta los puntos interiores y frontera (valores máximos y mínimos locales) y evaluar la función en dichos puntos. Finalmente, se debe escoger el máximo y el mínimo de todos ellos, ya que los extremos absolutos son también máximos y mínimos locales.

Ejemplo: Obtener los valores máximos y mínimos absolutos de:
$f(x,y) = 2 + 2x + 2y - x^2 - y^2$, en la región triangular acotada por las rectas $x = 0$, $y = 0$, $y = 9 - x$

Gráfica de la función y la región acotada:

```
> restart:
> with(plots):
> f:=(x,y)->2+2*x+2*y-x^2-y^2;
> FF:=plot3d(f(x,y),x=-0.5..9.5,y=-0.5..9.5,axes=frame,
  transparency=0.5):
> c1 := spacecurve([x, 0, 0], x = 0 .. 9, color = blue,
  thickness = 5):
> c2 := spacecurve([0, y, 0], y = 0 .. 9, color = blue,
  thickness = 5):
> c3 := spacecurve([x, 9-x, 0], x = 0 .. 9, color = blue,
```

```
   thickness = 5):
> cc1 := spacecurve([x, 0, f(x, 0)], x = 0 .. 9, color = red,
   thickness = 5):
> cc2 := spacecurve([0, y, f(0, y)], y = 0 .. 9, color = red,
   thickness = 5):
> cc3 := spacecurve([x, 9-x, f(x, 9-x)], x = 0 .. 9, color =
   red, thickness = 5):
> display3d([FF,c1,c2,c3,cc1,cc2,cc3], axes = normal);
```

Puntos interiores:

Primero se calculan los puntos extremos interiores de igual forma como en los extremos relativos.

```
> with(Student[VectorCalculus]):
> J:=Jacobian([f(x,y)],[x,y]);
> ptos:=solve({J[1,1]=0, J[1,2]=0}, {x,y});
> H := Hessian(f(x, y), [x, y]);
> with(Student[LinearAlgebra]):
> eval(Determinant(H), ptos);
```

Se guarda el punto crítico obtenido en el vector `PtosCrit`:

```
> PtosCrit[1]:=subs([ptos[1],ptos[2]],[x,y,f(x, y)]);
```

¿Cuáles son los puntos críticos obtenidos?

Puntos frontera:

Se toma un lado a la vez de la región triángular que acota la función:

Lado OA: Se evalua $f(x, y)$ en la recta $y = 0$, y se buscan los puntos críticos allí

```
> f1:= x->f(x,0);
> ptos:=solve(diff(f1(x),x)=0,x);
```

Guardamos los puntos críticos encontrados más los de frontera de nuevo en el vector `PtosCrit`. En este caso, el nuevo punto es el $(1, 0)$ cuya imagen es 3, mientras que los valores frontera son $(0, 0)$ y $(9, 0)$.

```
> PtosCrit[2]:=[ptos,0,f(ptos,0)];
> PtosCrit[3]:=[0,0,f(0,0)];
```

```
> PtosCrit[4]:=[9,0,f(9,0)];
```

Lado OB: Se evalua $f(x, y)$ en la recta $x = 0$, y se buscan los puntos críticos allí

```
> f2:= x->f(0,y);
> ptos:=solve(diff(f2(y),y)=0,y);
```

Guardamos los puntos críticos encontrados más los de frontera (el punto (0,0,2) ya está guardado del paso anterior)

```
> PtosCrit[5]:=[0,ptos,f(0,ptos)];
> PtosCrit[6]:=[0,9,f(0,9)];
```

Lado AB: Se evalua $f(x, y)$ en la recta $y = 9 - x$, y se buscan los puntos críticos allí

```
> f3:= x->f(x,9-x);
> ptos:=solve(diff(f3(x),x)=0,x);
```

Guardamos los puntos críticos encontrados más los de frontera (los puntos (0,9,-61) y (9,0,-61) ya están guardados de los pasos anteriores)

```
> PtosCrit[7]:=[ptos,9-ptos,f(ptos,9-ptos)];
```

Teniendo en cuenta todos los puntos críticos encontrados, se observa que el máximo es 4, y $f(x, y)$ lo asume en (1,1). El mínimo es -61, y $f(x, y)$ lo asume en (0,9) y (9,0).

```
> seq(PtosCrit[i], i = 1 .. 7);
```

Se representa gráficamente la función, la región acotada y los puntos para comprobar los resultados:

```
> pp:=pointplot3d({seq(PtosCrit[i], i = 1 .. 7)}, symbol =
  solidcircle, color = blue,symbolsize=30):
> display3d([FF,cc1,cc2,cc3,pp], axes = normal);
```

Extremos condicionados - Multiplicadores de Lagrange

Algunas veces es necesario obtener los valores extremos de una función $z = f(x, y)$ cuyo

dominio está restringido a cierto subconjunto particular del plano. El **método de los multiplicadores de Lagrange** reduce el problema restringido a uno sin restricciones cuyas ecuaciones pueden ser resueltas más fácilmente. De un espacio original de n variables pasa a uno de $n+1$ variables. Estas nuevas variables escalares desconocidas, una para cada restricción, son llamadas multiplicadores de Lagrange. El método dice que buscar los extremos condicionados de una función con un número determinado restricciones, por ejemplo k, será equivalente a buscar los extremos sin restricciones de una nueva función construida como una combinación lineal de la función y las restricciones, donde los coeficientes de las restricciones son los multiplicadores. Para encontrar estos valores de $f(x, y, z)$, cuyas variables están sujetas a la restricción $g(x, y, z) = 0$, sobre la superficie $g = 0$, se encuentran los puntos donde: $\nabla f = \lambda \nabla g$ para algún escalar λ (llamado multiplicador de Lagrange).
Ahora bien, suponiendo que $f(x, y, z)$, $g(x, y, z) = 0$, son derivables y que $\nabla g = 0$ cuando $g(x, y, z) = 0$. Los valores máximos y mínimos de f sujeta a la restricción $g(x, y, z) = 0$, se determinan a partir de obtener los valores x, y, z y λ que satisfagan en forma simultánea las ecuaciones: $\nabla f = \lambda \nabla g$ y $g(x, y, z) = 0$. En caso de funciones de dos variables independientes, la condición es similar, pero sin la variable z.

De forma más precisa, dado el problema de extremos condicionados

$$\text{extremar } f(x, y) \text{ sujeto a la condición } g(x, y) = 0$$

la regla de cálculo para resolver este problema es:

Paso 1. Definir la función $\phi(x, y, \lambda) = f(x, y) - \lambda g(x, y)$.
Paso 2. Resolver $\nabla \phi = 0$, en función de x, y y λ.
Paso 3. En los valores que resuelven el paso 2, evaluar f. Los de mayor imagen por f serán posibles máximos condicionados, y los de menor imagen por f serán posibles mínimos condicionados.

Ejemplo: Encontrar el máximo y el mínimo de la función $f(x, y) = x^2 + 4y^3$ bajo la condición $x^2 + 2y^2 = 1$ aplicando el método de los multiplicadores de Lagrange. Para lograrlo, primero se seguiran los pasos necesarios para la resolución a mano, utilizando el comando **Jacobian** y **Hessian**, para finalmente resolverlo y comparar las respuestas obtenidas utilizando la instrucción **extrema**.

```
> restart:
> with(plots):
> f:=(x,y)->x^2+4*y^3;
> g:=(x,y)->x^2+2*y^2-1;
> F:=plot3d(f(x,y),x=-1.1..1.1,y=-1.1..1.1,color="Red",
```

```
  transparency=0.6):
> G:=plot3d(g(x,y),x=-1.1..1.1,y=-1.1..1.1,color=blue,
  transparency=0.6):
> a1 := spacecurve([cos(t), sin(t)/sqrt(2), 0], t = 0 .. 2*Pi,
  color = blue, thickness = 4):
> a2 := spacecurve([cos(t), sin(t)/sqrt(2), f(cos(t), sin(t)
  /sqrt(2))], t = 0 .. 2*Pi, color = red, thickness = 4):
> display3d([F,a2], axes = frame, orientation = [-30, 80]);
```

A continuación, se construye la función auxiliar y se buscan los puntos críticos utilizando la libreria **Student[VectorCalculus]**, donde se encuentran los comandos: **Jacobian**, **Hessian** y **extrema**.

```
> with(Student[VectorCalculus]):
> phi := f(x,y)+lambda*g(x,y);
> J := Jacobian([phi], [x, y, lambda]);
> sol := solve({J[1, 1] = 0, J[1, 2] = 0, J[1, 3] = 0},
  {lambda, x, y});
> sol := seq(allvalues(sol[i]), i = 1 .. 4);
```

A la gráfica obtenida anteriormente se añaden los puntos críticos:

```
> for i to 6 do p[i] := subs(sol[i], [x, y, f(x, y)]) end do;
> a3 := pointplot3d({seq(p[i], i = 1 .. 6)}, symbol = box,
  color = black):
> display3d([a3,F,a2], axes = frame, orientation = [Student
  [VectorCalculus][`-`](30), 80]);
> display3d([a2, F], axes = frame, orientation = [-30, 80]);
> H:=Hessian(phi, [x, y, lambda]);
> with(Student[LinearAlgebra]):
> for i from 1 to 6 do q[i]:=eval(Determinant(H),sol[i])od;
```

Por último se calcula el valor de la función en los puntos críticos utilizando el comando **extrema**. Observe que con solo una línea se obtienen los extremos:

```
> extrema(f(x,y),g(x,y),{x,y},'s');
```

```
> s;
```

Ejercicios de autoavaluación

<u>Ejercicio propuesto 1:</u> **Calcular los puntos de la superficie** $x^2 + y^2 = z + 2$ **que se encuentran más cercanos al origen.**

Observación: La función distancia (euclídea) al origen en $\mathbb{R}^3$ **es:** $d = \sqrt{x^2 + y^2 + z^2}$

Para resolver este ejercicio primero se debe decidir qué función es la que se extrema y qué función es la ligadura.

```
> restart:
> with(plots):
> f:=(x,y,z)->sqrt(x^2+y^2+z^2);
> g:=(x,y,z)->x^2+y^2-z-2;
```

Una vez determinadas las dos funciones, vamos a representar gráficamente la función ligadura:

```
> G:=implicitplot3d(g(x,y,z),x=-3..3,y=-3..3,z=-3..1,color=
  blue,transparency=0.9,numpoints=20000):
> display3d(G,axes=normal,orientation=[-30,80]);
```

Utilizando directamente el comando extrema, se obtienen los valores extremos:

```
> extrema(f(x,y,z),g(x,y,z),{x,y,z},'s');
> s;
```

Se grafican sobre la curva anterior:

```
> for i from 1 to 5 do p[i]:=subs(s[i],[x,y,z]) end do;
> a := pointplot3d({p[1], p[4], p[5]}, symbol = box, color =
  red):
> cur1 := plot3d(p[2], x = -sqrt(6)/2 .. sqrt(6)/2, y = 0 ..
```

```
sqrt(6)/2, color = red, thickness = 4,numpoints=2000):
> cur2 := plot3d(p[3], x = -sqrt(6)/2 .. sqrt(6)/2, y = -sqrt
(6)/2..0, color = red, thickness = 4):
> display3d([a, cur1, cur2, G], axes = normal, orientation =
[-30, 80]);
```

Ejercicio propuesto 2:

Para la función $(x, y) = (x-3)^2 + y^2$ calcular los extremos relativos y absolutos.

Nuevamente utilizando los multiplicadores de Lagrange y el comando extrema.

```
> restart:
> with(plots):
> with(LinearAlgebra):
> f:=(x,y)->(x-3)^(2)+y^(2);
> plot3d(f(x, y), x = -10 .. 16, y = -10 .. 10, axes = normal);
```

Usando "extrema":

```
> extrema(f(x, y),{}, {x, y}, 's');
> s;
```

Para decidir cuáles de estos puntos son extremos relativos de $f(x, y)$, se evalúa la matriz hessiana en dichos puntos y se analiza el signo:

```
> with(VectorCalculus):
> J:=Jacobian([f(x,y)],[x,y]);
> solve({J[1,1]=0,J[1,2]=0},{x,y});
> Hessian(f(x,y),[x,y]);
```

Modificando el problema anterior para encontrar los máximos y los mínimos que se encuentran dentro del conjunto del plano $A = \{(x, y) \in \mathbb{R}^2 \mid x^2 + y^2 = 2\}$. Los extremos que encuentren en A se denominan extremos absolutos. Es decir, dentro de A no existirá ningún otro valor con imagen mayor a la del máximo ni menor a la del mínimo absoluto.

```
> plot3d(x^2+y^2, x = -2 .. 2, y = -2 .. 2, axes = normal):
```

Ejercicio propuesto 3: Calcular los valores extremos locales de la función $f(x, y) = 3y^2 - 2y^3 - 3x^2 + 6xy$, observe que la función tiene un punto de silla en el origen .

Exercicios propuestos complementarios

Ejercicio propuesto 4: Para resolver problemas de valores extremos con restricciones algebraicas sobre las variables se requiere por lo general del método de multiplicadores de Lagrange. Sin embargo en algunas ocasiones, es posible resolver estos problemas de manera directa, como en el siguiente problema:

Una compañía de mensajería sólo acepta cajas rectangulares en las que la suma de su largo y su circunferencia (perímetro de una sección transversal) no exceda 180 cm. Determinar las dimensiones de una caja aceptable con el volumen máximo.

Observación: Tener en cuenta que x, y, z, definan, el largo, ancho y la altura de la caja respectivamente. El volumen de la caja es $V = xyz$ con la condición $x + 2y + 2z = 108$ (la caja más grande que acepta la compañía de mensajería).

$$V = xyz$$
$$x = 108 - 2y - 2z$$
$$V(y, z) = (108 - 2y - 2z)\,yz$$
$$V(y, z) = 108yz - 2y^2z - 2yz^2$$

Ejercicio propuesto 5: Encontrar los valores extremos de $f(x, y) = xy$ sujetos a la restricción $g(x, y) = \frac{x^2}{8} + \frac{y^2}{2} - 1 = 0$